
Stat 342  Lab Week 10 – proc glm, proc freq 
For this lab, you will need the cereal.csv dataset in the email. 

Dataset source: Carl Schwarz's online notes 

Textbook source: R and SAS Chapter 5-6. 

Additional sources: http://analytics.ncsu.edu/sesug/2011/PO04.Hirabayashi.pdf  for plotting. 

  http://www.littledumbdoctor.com/index_files/sas_3_proc_glm.pdf for regression 

 

Part I, Importing. 
 
proc import 
datafile='dir_location\cereal.csv' 
 out=cereal dbms=csv; 

 delimiter=','; 
 getnames=yes; 
run; 

 

 

Part II, Regression on numeric variables. 
 

In a general linear models procedure, PROC GLM, make a model of sugars as a function of protein, fat, 

and sodium. Note that 'sugars', 'protein', 'fat', and 'sodium' are all numeric. 

Use the option 'p' to get regression coefficients, predictions, and residuals. 

Use the option 'clm' to get confidence limits of the mean response. 

 

What is the regression equation? 

 
 

proc glm data = cereal; 
 model sugars= protein fat sodium / p clm; 
run; 

 

 

Save the results using an output statement without settings. What information gets saved? 

 
 

proc glm data = cereal; 
 model sugars= protein fat sodium / p clm; 
 output out = cereal_output; 
run; 

 

proc print data = cereal_output; 
run; 

 

 

 

 

 

 

http://analytics.ncsu.edu/sesug/2011/PO04.Hirabayashi.pdf
http://www.littledumbdoctor.com/index_files/sas_3_proc_glm.pdf


What happens to the saved dataset if you instead specify the four additional variables from the model? 

 
 

proc glm data = cereal; 
 model sugars= protein fat sodium / p clm; 
 output out = cereal_output P=predict R=resid L95=lwr U95=upr; 

run; 
 

proc print data = cereal_output; 
run; 

 

 

Compare the predicted amount of sugar to the actual amount of sugar. How closely do they compare. 

 
 

proc sgplot data=cereal_output; 
 scatter x=sugars y=predict; 
 title1 "Actual Sugars (x) vs. Predicted Sugars (y) in Cereal"; 
run; 

 

Use the residuals and the datalabels option to identify cereals that have much more or much less sugar 

than your model predicts. 

 
 

proc sgplot data=cereal_output; 
 scatter x=predict y=resid / datalabel = name; 

 title1 "Residuals (y) vs predicted sugar (x) in Cereal"; 
run; 
 

 

 

 

Part II, Regression on numeric and categorical variables. 

 
Now introduce a categorical variable – manufacturer – to the model. 

First, take a side-by-side boxplot. Part of regression is estimating the variance of responses. Can you 

see any manufacturers that might produce difficulties in estimating variance? 
 

proc sgplot data=cereal; 
  vbox sugars / category=manufacturer; 
run; 

 

 

 

 

 

 

 

 



 Use the 'solution' option in proc glm to get a regression equation predicting sugars using dummy 

variables to represent different manufacturers. 

Also, use the 'clparm' option to get confidence limits of these regression parameters. 
 

 

proc glm data = cereal; 
  class manufacturer; 

 model sugars= manufacturer protein fat sodium / solution clparm; 
run; 
 

Part III, Contingency tables and analysis of categorical variables. 

 
Cross-frequency tables (aka crosstabs, contingency tables) are a method of analyzing relationships 

between categorical variables. They also aid in hypothesis tests of association, such as chi-squared and 

odds ratio. 

 

First, look at the frequency of tables of each categorical variable individually. 

 
proc freq data=cereal; 
 tables manufacturer shelf type; 

run; 
 

 

Next, look at the crosstabs of manufacturer and shelf. Run each of these following frequency 

procedures separately – once with default settings... 
 

proc freq data=cereal; 
 tables manufacturer*shelf; 
run; 

 

...and once with much less cluttered settings... 
 

proc freq data=cereal; 
 tables manufacturer*shelf / nocol norow nopercent; 

run; 
 

 

 

Next, run some hypothesis tests to look for an association between maker and shelf. Are there some 

brands of cereal that always end up on the top shelf? 

 

The null hypothesis of the chi-squared test is that the two categorical variables are independent. Do we 

reject that hypothesis? Are there any problems with the chi-squared test being reported? 
 

proc freq data=cereal; 

 tables manufacturer*shelf / chisq nocol norow nopercent; 
run; 
 



 

When some of the cell sizes are small, one alternative to the chi-squared test is Fisher's exact test. 

Unfortunately, its combinatorial nature makes it very hard to compute. Instead, we can use the Monte-

Carlo alternative, which approximates the exact test. 

 

Do we reject the null hypothesis of independence now? 

 
proc freq data=cereal; 
 tables manufacturer*shelf; 
 exact fisher / mc n=1000; /*Do 10,000 monte carlo simulations*/ 

run; 
 

 

Finally, let's look at a three-way table. Try it in two different orders, because SAS will organize the 

table differently depending on the order. It will print a series of two-way tables for each value of the 

first variable listed. 

 
proc freq data=cereal; 

 tables manufacturer*shelf*type / nocol norow nopercent; 
run; 
 

proc freq data=cereal; 

 tables type*shelf* manufacturer / nocol norow nopercent; 
run; 
 

 

We can test if there is a relationship between shelf and cereal type (hot or cold), controlling for 

manufacturer with Cochran-Mantel-Haenszel (CMH) test, but we need to simplify shelf down to a 

binary variable first. This is done with a data step. 

 
 

data cereal; 
 set cereal; 

 top_shelf = "Other"; 
 if shelf eq "Top" then top_shelf = "Top"; 
run; 

 

proc freq data=cereal; 
 tables type*top_shelf* manufacturer / cmh nocol norow nopercent; 
run; 

 

 

 

 

 

 

 

 

 



Part IV, Polynomial terms, transformed variables. 

 
Sometimes the relationship between one variable and another can’t be explained by a line. In cases like 

this, we may wish to employ two more terms composed of the same variable. Say, for example, the 

saltiness and fattiness of different cereals followed a polynomial curve of order 3. 

 

To set up such a polynomial, use an interaction of an explanatory variable with itself. 

 
proc glm data = cereal; 
 model sodium = fat fat*fat fat*fat*fat / solution; 

run; 

 

Does this work with categorical variables as well? 

 
proc glm data = cereal; 

 class shelf; 
 model sodium = shelf shelf*shelf / solution; 

run; 

 

What happens when a new variable is introduced? Specifically, what plots are available when you are 

working with two different numeric variables, or three? 

 
proc glm data = cereal; 
 model sodium = fat fat*fat sugars / solution; 

run; 

 
proc glm data = cereal; 
 model sodium = fat fat*fat sugars protein / solution; 
run; 

 

 

Finally, what happens when we compute the transformed variables beforehand, and use those in place 

of the polynomial terms? 

 
data cereal; 

set cereal; 

fat2 = fat*fat; 

fat3 = fat*fat*fat; 

run; 

 

proc glm data = cereal; 

 model sodium = fat fat2 fat3 / solution; 
run; 

 
 

 

 

 



Part V, model selection. 

 
If there are several possible terms we can use to predict a model, we may wish to employ an automated 

method to find the ‘best’ such model. To do this, we need proc glmselect, a starting model (usually 

with all the possible terms to consider), a selection method, and a criterion for deciding how ‘good’ a 

model is. 

 

Try this predicting sodium using a bunch of different variables, the stepwise selection method, and the 

Akaike Information Criterion. 

 
proc glm data = cereal; 
 class shelf manufacturer; 

 model sodium = shelf manufacturer fat*fat fat sugars protein 

sugars*protein protein*shelf / selection=stepwise(select=AIC); 

run; 

 

 

 

These are not the only options. Notice that other criteria that are offered may find a different solution as 

the ‘best’ one. Try finding the ‘best’ model using the LASSO method (which is much more preferred in 

modern statistics), and using the Swartz-Bayesian Criterion (SBC, also called the Bayesian Information 

Criterion, BIC in R). 

 

  
proc glm data = cereal; 

 class shelf manufacturer; 

 model sodium = shelf manufacturer fat*fat fat sugars protein 

sugars*protein protein*shelf / selection=lasso(select=SBC); 

run; 

 

 


